

Verifpal analysis of tls13.vp

September 7, 2026

Abstract

This report describes the analysis of the Verifpal model `tls13.vp` against an active attacker, with each principal running 2 concurrent sessions. Of its 19 queries, 3 were broken. Every attack reported here is a witness against the model as written. A query reported as holding means that this search found no attack at these parameters, which is not a proof that none exists.

1 `tls13.vp`

Analysed against an active attacker at 2 sessions per principal, in 24755743 ms. The result code is `c1c0c0c0c0c0e0c0c0c1c0c1a0a0a0a0f0f0`.

1.1 Protocol

1.2 Verdicts

Table 1: Every query in the model, with the verdict this run reached.

Query
confidentiality? <i>shstraffic</i>
confidentiality? <i>shstraffic</i> [precondition[Server $\rightarrow$ Client: <i>appdata2</i>]]
confidentiality? <i>chstraffic_c</i> [precondition[Client $\rightarrow$ Server: <i>clientflight</i>]]
confidentiality? <i>saptraffice</i> [precondition[Server $\rightarrow$ Client: <i>appdata2</i>]]
confidentiality? <i>captraffice_c</i> [precondition[Client $\rightarrow$ Server: <i>clientflight</i>]]
confidentiality? <i>exportersecret</i> [precondition[Server $\rightarrow$ Client: <i>appdata2</i>]]
equivalence? <i>saptraffice</i> , <i>saptraffice_c</i> [precondition[Server $\rightarrow$ Client: <i>appdata2</i>] precondition[Client $\rightarrow$ Server: <i>clientflight</i>]]
confidentiality? <i>appmsg1</i>
confidentiality? <i>appmsg2</i>
confidentiality? <i>appmsg3</i>
confidentiality? <i>respsk2</i>
confidentiality? <i>clientid</i>
authentication? Server $\rightarrow$ Client: <i>serverflight</i>
authentication? Client $\rightarrow$ Server: <i>clientflight</i>
authentication? Client $\rightarrow$ Server: <i>appdata1</i>
authentication? Server $\rightarrow$ Client: <i>appdata2</i>
authentication? Server $\rightarrow$ Client: <i>appdata3</i>
freshness? <i>saptraffice</i>
freshness? <i>captraffice_c</i>



Figure 1: The protocol as written. Guarded values are bracketed; a dagger marks every value an attack below substitutes or replays.

1.3 Attacks

1.3.1 confidentiality? *shstraffic*

shstraffic ($\text{HKDF}(\text{HKDF}(\text{derivedes}, \text{DH}_{\text{KEX}}(\text{PUBKEY}(\text{nil}), \text{se}), \text{nil})_1, \text{labelshstraffic}, \text{HASH}(\text{PUBKEY}(\text{nil}), \text{gse}))|1)$ is obtained by Attacker.




Figure 2: How the attacker breaks confidentiality? *shstraffic*. Substituted values carry a dagger.

- ① Attacker constructs $\text{PUBKEY}(\text{nil})$.
- ② Attacker replaces *gce* (sent by Client to Server) with $\text{PUBKEY}(\text{nil})$. (*gce* was $\text{PUBKEY}(\text{ce})$) Client → Server
- ③ During Ca's run (phase 0), attacker constructs *earlysecret*.
- ④ During Ca's run (phase 0), attacker constructs *derivedes*.
- ⑤ Attacker observes *gse* on the wire.
- ⑥ Attacker constructs $\text{DH}_{\text{KEX}}(\text{gse}, \text{nil})$.
- ⑦ Attacker constructs $\text{HKDF}(\text{derivedes}, \text{DH}_{\text{KEX}}(\text{gse}, \text{nil}), \text{nil})|1$.
- ⑧ Attacker constructs $\text{HASH}(\text{PUBKEY}(\text{nil}), \text{gse})$.
- ⑨ Attacker constructs $\text{HKDF}(\text{HKDF}(\text{derivedes}, \text{DH}_{\text{KEX}}(\text{gse}, \text{nil}), \text{nil})_1, \text{labelchstraffic}, \text{HASH}(\text{PUBKEY}(\text{nil}), \text{gse}))|1$.
- ⑩ Attacker constructs $\text{HKDF}(\text{HKDF}(\text{HKDF}(\text{derivedes}, \text{DH}_{\text{KEX}}(\text{gse}, \text{nil}), \text{nil})_1, \text{labelchstraffic}, \text{HASH}(\text{PUBKEY}(\text{nil}), \text{gse}))|1, \text{labeliv}, \text{nil})|1$.
- ⑪ Attacker replaces *clientflight* (sent by Client to Server) with $\text{AEAD}_{\text{ENC}}(\text{HKDF}(\text{HKDF}(\text{derivedes}, \text{DH}_{\text{KEX}}(\text{gse}, \text{nil}), \text{nil})_1, \text{labelchstraffic}, \text{HASH}(\text{PUBKEY}(\text{nil}), \text{gse}))|1, \text{HASH}(\text{HKDF}(\text{HKDF}(\text{HKDF}(\text{derivedes}, \text{DH}_{\text{KEX}}(\text{gse}, \text{nil}), \text{nil})_1, \text{labelchstraffic}, \text{HASH}(\text{PUBKEY}(\text{nil}), \text{gse}))|1, \text{labeliv}, \text{nil})_1, \text{seq0}), \text{nil}, \text{recordheader})$. (*clientflight* was $\text{AEAD}_{\text{ENC}}(\text{chstraffic_c}, \text{HASH}(\text{chsiv_c}, \text{seq0}), \text{CONCAT}(\text{clientid}, \text{clientpk}, \text{clientcertsig}, \text{clientcv}, \text{clientfinished}), \text{recordheader})$) Client → Server
- ⑫ Server's $\text{AEAD}_{\text{DEC}}(\text{HKDF}(\text{HKDF}(\text{derivedes}, \text{DH}_{\text{KEX}}(\text{PUBKEY}(\text{nil}), \text{se}), \text{nil})_1, \text{labelchstraffic}, \text{HASH}(\text{PUBKEY}(\text{nil}), \text{gse}))|1, \text{HASH}(\text{HKDF}(\text{HKDF}(\text{HKDF}(\text{derivedes}, \text{DH}_{\text{KEX}}(\text{PUBKEY}(\text{nil}), \text{se}), \text{nil})_1, \text{labelchstraffic}, \text{HASH}(\text{PUBKEY}(\text{nil}), \text{gse}))|1, \text{labeliv}, \text{nil})_1, \text{seq0}), \text{AEAD}_{\text{ENC}}(\text{HKDF}(\text{HKDF}(\text{HKDF}(\text{derivedes}, \text{DH}_{\text{KEX}}(\text{PUBKEY}(\text{nil}), \text{se}), \text{nil})_1, \text{labelchstraffic}, \text{HASH}(\text{PUBKEY}(\text{nil}), \text{gse}))|1, \text{labeliv}, \text{nil})_1, \text{seq0}), \text{nil}, \text{recordheader}), \text{recordheader})$ passes — the attacker controls one of its inputs.
- ⑬ Attacker constructs $\text{HKDF}(\text{HKDF}(\text{derivedes}, \text{DH}_{\text{KEX}}(\text{gse}, \text{nil}), \text{nil})_1, \text{labelshstraffic}, \text{HASH}(\text{PUBKEY}(\text{nil}), \text{gse}))|1$.

1.3.2 confidentiality? *appmsg3*

appmsg3 (*appmsg3*) is obtained by Attacker.



Figure 3: How the attacker breaks confidentiality? *appmsg3*. Substituted values carry a dagger.

- ① Attacker observes *appdata3* on the wire.
- ② Attacker is handed *saptraffic1* by a leaks declaration.

- ③ Attacker constructs *sapiv1*.
- ④ Attacker constructs $\text{HASH}(\text{sapiv1}, \text{seq0})$.
- ⑤ Attacker opens *appdata3* with *saptraffic1*, $\text{HASH}(\text{sapiv1}, \text{seq0})$, obtaining *appmsg3*.

1.3.3 confidentiality? *clientid*

clientid (*clientid*) is obtained by Attacker.




Figure 4: How the attacker breaks confidentiality? *clientid*. Substituted values carry a dagger.

- ① Attacker constructs $\text{PUBKEY}(\text{nil})$.
 - ② During Ca's run (phase 0), attacker constructs *earlysecret*.
 - ③ During Ca's run (phase 0), attacker constructs *derivedes*.
 - ④ Attacker observes gce@2 on the wire.
 - ⑤ Attacker constructs $\text{DH}_{\text{KEX}}(\text{gce}^{\text{a2}}, \text{nil})$.
 - ⑥ Attacker constructs $\text{HKDF}(\text{derivedes}, \text{DH}_{\text{KEX}}(\text{gce}^{\text{a2}}, \text{nil}), \text{nil})|1$.
 - ⑦ Attacker constructs $\text{HASH}(\text{gce}^{\text{a2}}, \text{PUBKEY}(\text{nil}))$.
 - ⑧ Attacker constructs $\text{HKDF}(\text{HKDF}(\text{derivedes}, \text{DH}_{\text{KEX}}(\text{gce}^{\text{a2}}, \text{nil}), \text{nil})_1, \text{labelshstraffic}, \text{HASH}(\text{gce}^{\text{a2}}, \text{PUBKEY}(\text{nil})))|1$.
 - ⑨ Attacker constructs $\text{HKDF}(\text{HKDF}(\text{HKDF}(\text{derivedes}, \text{DH}_{\text{KEX}}(\text{gce}^{\text{a2}}, \text{nil}), \text{nil})_1, \text{labelshstraffic}, \text{HASH}(\text{gce}^{\text{a2}}, \text{PUBKEY}(\text{nil})))_1, \text{labeliv}, \text{nil})|1$.
 - ⑩ During Client@2's run (phase 0), attacker constructs $\text{HASH}(\text{HKDF}(\text{HKDF}(\text{HKDF}(\text{derivedes}, \text{DH}_{\text{KEX}}(\text{gce}^{\text{a2}}, \text{nil}), \text{nil})_1, \text{labelshstraffic}, \text{HASH}(\text{gce}^{\text{a2}}, \text{PUBKEY}(\text{nil})))_1, \text{labeliv}, \text{nil})_1, \text{seq0})$.
 - ⑪ Attacker observes *gmsigkey* on the wire.
 - ⑫ Attacker observes *mallcertsig* on the wire.
 - ⑬ During Client@2's run (phase 0), attacker constructs $\text{HKDF}(\text{HKDF}(\text{HKDF}(\text{derivedes}, \text{DH}_{\text{KEX}}(\text{gce}^{\text{a2}}, \text{nil}), \text{nil})_1, \text{labelshstraffic}, \text{HASH}(\text{gce}^{\text{a2}}, \text{PUBKEY}(\text{nil})))_1, \text{labelfinished}, \text{nil})|1$.
 - ⑭ During Client@2's run (phase 0), attacker constructs $\text{HASH}(\text{HASH}(\text{gce}^{\text{a2}}, \text{PUBKEY}(\text{nil})), \text{nil}, \text{gmsigkey}, \text{mallcertsig})$.
 - ⑮ During Client@2's run (phase 0), attacker constructs $\text{HASH}(\text{HASH}(\text{HASH}(\text{gce}^{\text{a2}}, \text{PUBKEY}(\text{nil})), \text{nil}, \text{gmsigkey}, \text{mallcertsig}), \text{nil})$.
 - ⑯ During Client@2's run (phase 0), attacker constructs $\text{MAC}(\text{HKDF}(\text{HKDF}(\text{HKDF}(\text{derivedes}, \text{DH}_{\text{KEX}}(\text{gce}^{\text{a2}}, \text{nil}), \text{nil})_1, \text{labelshstraffic}, \text{HASH}(\text{gce}^{\text{a2}}, \text{PUBKEY}(\text{nil})))_1, \text{labelfinished}, \text{nil})_1, \text{HASH}(\text{HASH}(\text{HASH}(\text{gce}^{\text{a2}}, \text{PUBKEY}(\text{nil})), \text{nil}, \text{gmsigkey}, \text{mallcertsig}), \text{nil}))$.
 - ⑰ During Client@2's run (phase 0), attacker constructs $\text{CONCAT}(\text{nil}, \text{gmsigkey}, \text{mallcertsig}, \text{nil}, \text{MAC}(\text{HKDF}(\text{HKDF}(\text{HKDF}(\text{derivedes}, \text{DH}_{\text{KEX}}(\text{gce}^{\text{a2}}, \text{nil}), \text{nil})_1, \text{labelshstraffic}, \text{HASH}(\text{gce}^{\text{a2}}, \text{PUBKEY}(\text{nil})))_1, \text{labelfinished}, \text{nil})_1, \text{HASH}(\text{HASH}(\text{HASH}(\text{gce}^{\text{a2}}, \text{PUBKEY}(\text{nil})), \text{nil}, \text{gmsigkey}, \text{mallcertsig}), \text{nil})))$.
 - ⑱ During Client@2's run (phase 0), attacker constructs $\text{AEAD}_{\text{ENC}}(\text{HKDF}(\text{HKDF}(\text{derivedes}, \text{DH}_{\text{KEX}}(\text{gce}^{\text{a2}}, \text{nil}), \text{nil})_1, \text{labelshstraffic}, \text{HASH}(\text{gce}^{\text{a2}}, \text{PUBKEY}(\text{nil})))_1, \text{HASH}(\text{HKDF}(\text{HKDF}(\text{HKDF}(\text{derivedes}, \text{DH}_{\text{KEX}}(\text{gce}^{\text{a2}}, \text{nil}), \text{nil})_1, \text{labelshstraffic}, \text{HASH}(\text{gce}^{\text{a2}}, \text{PUBKEY}(\text{nil})))_1, \text{labeliv}, \text{nil})_1, \text{seq0}), \text{CONCAT}(\text{nil}, \text{gmsigkey}, \text{mallcertsig}, \text{nil}, \text{MAC}(\text{HKDF}(\text{HKDF}(\text{HKDF}(\text{derivedes}, \text{DH}_{\text{KEX}}(\text{gce}^{\text{a2}}, \text{nil}), \text{nil})_1, \text{labelshstraffic}, \text{HASH}(\text{gce}^{\text{a2}}, \text{PUBKEY}(\text{nil})))_1, \text{labelfinished}, \text{nil})_1, \text{HASH}(\text{HASH}(\text{HASH}(\text{gce}^{\text{a2}}, \text{PUBKEY}(\text{nil})), \text{nil}, \text{gmsigkey}, \text{mallcertsig}), \text{nil}))), \text{recordheader})$.
 - ⑲ Attacker replaces $\text{gse}^{\text{a2}}, \text{serverflight}^{\text{a2}}$ (sent by Server@2 to Client@2) with $\text{PUBKEY}(\text{nil}), \text{AEAD}_{\text{ENC}}(\text{HKDF}(\text{HKDF}(\text{derivedes}, \text{DH}_{\text{KEX}}(\text{gce}^{\text{a2}}, \text{nil}), \text{nil})_1, \text{labelshstraffic}, \text{HASH}(\text{gce}^{\text{a2}}, \text{PUBKEY}(\text{nil})))_1, \text{HASH}(\text{HKDF}(\text{HKDF}(\text{HKDF}(\text{derivedes}, \text{DH}_{\text{KEX}}(\text{gce}^{\text{a2}}, \text{nil}), \text{nil})_1, \text{labelshstraffic}, \text{HASH}(\text{gce}^{\text{a2}}, \text{PUBKEY}(\text{nil})))_1, \text{labeliv}, \text{nil})_1, \text{seq0}), \text{CONCAT}(\text{nil}, \text{gmsigkey}, \text{mallcertsig}, \text{nil}, \text{MAC}(\text{HKDF}(\text{HKDF}(\text{HKDF}(\text{derivedes}, \text{DH}_{\text{KEX}}(\text{gce}^{\text{a2}}, \text{nil}), \text{nil})_1, \text{labelshstraffic}, \text{HASH}(\text{gce}^{\text{a2}}, \text{PUBKEY}(\text{nil})))_1, \text{labelfinished}, \text{nil})_1, \text{HASH}(\text{HASH}(\text{HASH}(\text{gce}^{\text{a2}}, \text{PUBKEY}(\text{nil})), \text{nil}, \text{gmsigkey}, \text{mallcertsig}), \text{nil}))), \text{recordheader})$. (gse^{a2} was $\text{PUBKEY}(\text{se}^{\text{a2}})$; $\text{serverflight}^{\text{a2}}$ was $\text{AEAD}_{\text{ENC}}(\text{shstraffic}^{\text{a2}}, \text{HASH}(\text{shsiv}^{\text{a2}}, \text{seq0}), \text{CONCAT}(\text{servername}, \text{gsigkey}^{\text{a2}}, \text{servercertsig}^{\text{a2}}, \text{servercv}^{\text{a2}}, \text{serverfinished}^{\text{a2}}), \text{recordheader}))$
- Server@2 → Client@2

- [illegible]

1.4 Scope and assumptions

Every verdict above was reached against an active attacker, with each principal running 2 concurrent sessions, over exactly the model as written. An attack is a witness and stands on its own. A query reported as holding says only that this search found no attack at those parameters: the search space this engine defines was explored, which is never the space of all attacks.

- Scenario: Client runs with *peer*cert as *server*cert, a honest peer.
- Scenario: Client runs with *peer*cert as *mall*cert, a compromised peer.
- Per-scenario values and principals carry the suffix @2 onward.

A Model source

A.1 tls13.vp

```
1 // SPDX-FileCopyrightText: © 2019-2026 Nadim Kobeissi <nadim@symbolic.software>
2 // SPDX-License-Identifier: GPL-3.0-only
3 //
4 // TLS 1.3 (RFC 9846, which obsoletes RFC 8446): the full handshake of
5 // Figure 1 with certificate-based authentication of both endpoints,
6 // followed by the post-handshake messages of §4.7. The server
7 // authenticates with a certificate, the server requests one from the
8 // client and the client authenticates with it, both certificates are
9 // issued by a certificate authority, and once the handshake completes
10 // the server hands out two resumption tickets and rotates its sending
11 // key. The companion model tls13-Ortt.vp is the resumption handshake
12 // those tickets are for.
13 //
14 // What is here, section by section:
15 // §4.1 the running transcript hash, extended message by message;
16 // §4.2 ClientHello and ServerHello as key shares;
17 // §4.3.8 fresh key shares per connection (`generates ce`, `generates se`);
18 // §4.5.1 Certificate as (name, key, CA signature over both), for both
19 // endpoints, validated against a trust anchor received guarded;
20 // §4.5.2 CertificateVerify over the transcript, prefixed by the context
21 // string that separates a server's signature from a client's;
22 // §4.5.3 Finished as a MAC keyed by a finished key expanded from the
23 // sender's handshake traffic secret, over the transcript through
24 // CertificateVerify;
25 // §5.2 records as AEAD under the traffic secret of the sender, with
26 // the record header as the additional data;
27 // §5.3 the per-record nonce formed from the sender's write IV and the
28 // §7.3 record sequence number, the IV expanded from the traffic
29 // secret under its own label, so a record is accepted only in
30 // its place (F.2, order protection) and nothing about the nonce
31 // travels;
32 // §7.1 the key schedule from HKDF-Extract(0, 0) through the derived
33 // steps to the main secret, every Derive-Secret carrying its own
34 // label and the transcript hash of that point, plus the exporter
35 // and resumption secrets hanging off the main secret;
36 // §4.7.1 NewSessionTicket, with the PSK derived from the resumption
37 // secret and the ticket nonce, two tickets on one connection;
38 // §4.7.3 KeyUpdate under the old key, then §7.2's next-generation
39 // §7.2 application traffic secret for the record that follows.
40 //
41 // Expected: c1c0c0c0c0c0e0c0c1c0c1a0a0a0a0f0f0, at two sessions and
42 // at one. That code was derived by reading the protocol, not by running
```

```

43 // the model; confirm it before pinning it.
44 //
45 // The queries follow Appendix F, and three of them fail on purpose. Read
46 // each failure with the RFC open, because each one is a sentence the RFC
47 // itself wrote.
48 //
49 // Secrecy of the session keys (F.1), and CK01's proviso:
50 // confidentiality? sHsTraffic FAILS
51 // confidentiality? sHsTraffic [precondition: server completes] HOLDS
52 // confidentiality? cHsTraffic_c [precondition: client completes] HOLDS
53 // confidentiality? sApTraffic [precondition: server completes] HOLDS
54 // confidentiality? cApTraffic_c [precondition: client completes] HOLDS
55 // confidentiality? exporterSecret [precondition: server completes] HOLDS
56 // equivalence? sApTraffic, sApTraffic_c [both complete] HOLDS
57 // The server derives its handshake secrets from whatever key share
58 // arrives in the ClientHello, before anything has been checked, so an
59 // attacker that opens a connection with a share of its own owns a
60 // sHsTraffic. F.1 says as much: "the attacker can establish its own
61 // session keys with the server, but those session keys are distinct from
62 // those established by the client." What the server does with such a key
63 // is halt, at the client's CertificateVerify, without ever sending
64 // application data under it. The gated queries say exactly what F.1 says
65 // under "Establishing the same session keys": the guarantee holds
66 // "provided that it completes successfully on each endpoint". The
67 // precondition restricts each query to executions in which the named
68 // endpoint goes on past its last check, and in every such execution the
69 // keys are a two-party secret and the two sides hold the same one. The
70 // client has the same share on its side: it derives cHsTraffic_c before
71 // it can decrypt the server's flight, so the ungated client-side query
72 // would fail the same way, and the gated one holds.
73 //
74 // Application data (F.2 confidentiality, F.1 forward secrecy):
75 // confidentiality? appMsg1 HOLDS
76 // confidentiality? appMsg2 HOLDS
77 // confidentiality? appMsg3 FAILS
78 // Phase 1 leaks both long-term signing keys after the connection is over.
79 // Nothing recorded opens: the traffic secrets descend from the ephemeral
80 // dhe = DH_KEX(gce, se), and neither signing key is an ingredient. That is
81 // F.1's "forward secret with respect to long-term keys". Phase 2 leaks
82 // sApTraffic1, the server's sending secret after its KeyUpdate. appMsg3
83 // was encrypted under it and falls, the attacker expanding the write IV
84 // from the leaked secret and forming the nonce as the client does;
85 // appMsg2 was encrypted under the generation before, which
86 // HKDF-Expand-Label cannot be run backwards to, and holds. That is F.2's
87 // "forward secrecy after key change", and the only failure among the
88 // application data. F.2 also says the other direction is not provided:
89 // leak generation 0 instead and every later generation follows from it.
90 // This model does not show that, because one file can leak in one order.
91 //
92 // Resumption (F.1, tickets):
93 // confidentiality? resPsk2 HOLDS
94 // Phase 2 also leaks resPsk1, the PSK behind the first ticket. The second
95 // ticket's PSK is a separate expansion of the resumption secret under its
96 // own nonce, and survives, as does the resumption secret itself under the
97 // traffic-secret leak: "the resumption secret computed by connection N
98 // and needed to form connection N+1 is separate from the traffic keys
99 // used by connection N".
100 //
101 // Endpoint identities (F.1):
102 // confidentiality? clientId FAILS

```



```

103 // The client's identity travels only inside its Certificate, sealed under
104 // cHsTraffic_c after the server has authenticated, so a network attacker
105 // never sees it. It is disclosed all the same, and the reason is the
106 // second scenario: there the client is configured to reach Mallory, whose
107 // signing key the attacker holds, and the attacker completes the server
108 // side of that handshake with Mallory's genuine certificate. The client
109 // then does what F.1 promises it will do: "it only reveals its identity
110 // to an authenticated server". Mallory is one. Verifpal's confidentiality
111 // query asks whether the attacker knows the value, not whom the client
112 // meant to tell, and a `knows private` identity is the same value in
113 // every run. Delete the second scenario and this query is expected to
114 // hold. The server's identity is public here, as a hostname is; the RFC
115 // protects the server's certificate only against passive attackers, and
116 // an active one learns it by opening a connection.
117 //
118 // Peer authentication (F.1) and integrity with non-replay (F.2):
119 // authentication? Server → Client: serverFlight HOLDS
120 // authentication? Client → Server: clientFlight HOLDS
121 // authentication? Client → Server: appData1 HOLDS
122 // authentication? Server → Client: appData2 HOLDS
123 // authentication? Server → Client: appData3 HOLDS
124 // Verifpal's authentication query is injective agreement, so a hold rules
125 // out replay as well as forgery. The server's flight is accepted only
126 // under a signature, by a key a trust anchor vouches for, over a
127 // transcript that contains the client's own key share; the client's
128 // flight is accepted under the same construction, over a transcript that
129 // also contains the server's certificate and Finished. The second half is
130 // where the scenarios[] block earns its place. The client that talks to
131 // Mallory signs a CertificateVerify for that handshake, and the attacker
132 // holds it afterwards. Reusing it against the honest server is the attack
133 // F.1 describes for constructions "that do not bind to the server's
134 // certificate" ([PSK-FINISHED], [Kraw16]), and it fails here because the
135 // client's signature covers the server's Certificate and the shares of
136 // this connection. Every record after the handshake is sealed under a
137 // secret that mixes both ephemeral shares, so a record lifted from a
138 // concurrent connection decrypts nowhere, and the sequence number in the
139 // nonce keeps the four records the server sends under one key from
140 // standing in for each other.
141 //
142 // Uniqueness of the session keys (F.1):
143 // freshness? sApTraffic, freshness? cApTraffic_c HOLD
144 // Both traffic secrets descend from the per-connection se and ce.
145 //
146 // Modelling notes.
147 //
148 // The scenarios[] block gives the client two configurations: the
149 // certificate of the honest server, or Mallory's. In TLS that
150 // configuration is a hostname. It is a certificate here because Verifpal
151 // tells a corrupt peer from an honest one by following a leaked key into
152 // the values computed from it, and a bare name is computed from nothing:
153 // Mallory's certificate carries Mallory's public key, whose private half
154 // leaks, so a run configured with it is a run with a corrupt peer and its
155 // claims are not the protocol's to answer for. The client reads the
156 // expected name out of the configured certificate and checks the CA's
157 // signature over that name and the key in the Certificate message it
158 // receives, which is hostname verification as a browser does it; nothing
159 // else about the configured certificate is consulted. Check the received
160 // name instead and the attacker presents Mallory's genuine certificate to
161 // a client that wanted the honest server, which is the misbinding of F.9.
162 //

```



```

163 // The server's certificate check has the same shape, and needs it. The
164 // CA certifies Mallory too, so a server that accepted any certificate
165 // the CA issued would accept Mallory as its client, and the attacker,
166 // holding Mallory's key, would complete the client side of the handshake
167 // against the honest server: the client-flight and appData1
168 // authentication queries would fail and the server's gated session
169 // secrets would be shared with the attacker, all of it correct TLS with
170 // a peer the server never meant to serve. A deployment with client
171 // certificates decides which certified identities it serves, so the
172 // server here holds the identity of its one client and checks the CA's
173 // signature over that identity and the key in the certificate it
174 // receives. The transcript still covers the bytes as received (§4.1).
175 //
176 // Client certificates change the reading of this model against the
177 // server-only handshake that browsers run. Without them the server has no
178 // way to tell the client from an attacker with a key share, and every
179 // claim the server makes about who it talked to fails: authentication of
180 // the client's flight and data, and the confidentiality of the server's
181 // own session secrets, since "the client" may simply be the attacker.
182 // With them those claims hold, and the ungated sHsTraffic query above is
183 // what remains of that story.
184 //
185 // Records follow §5.2 and §5.3. Each traffic secret stands for the write
186 // key §7.3 expands from it, and the write IV is expanded from it under
187 // the "iv" label; the per-record nonce is that IV combined with the
188 // record sequence number, written HASH(iv, seq) because Verifpal has no
189 // XOR, and the additional data is the record header, the same public
190 // bytes on every record. Sequence numbers start over whenever the key
191 // changes (§5.3), so the first record under the updated secret is seq0
192 // again. An earlier version of this model drew a fresh nonce per record
193 // and sent it beside the ciphertext; TLS does neither, and the extra
194 // wire slot per record was what put the analysis out of reach.
195 //
196 // Three things the RFC has are left out because they cannot move a
197 // verdict and each one widened the search. The ClientHello and
198 // ServerHello randoms are absent: the transcript already binds the fresh
199 // key shares, and a random is a bare constant nothing checks on its own.
200 // EncryptedExtensions and CertificateRequest are absent from the
201 // transcript: as constants both sides already hold they could only sit in
202 // it inertly (parameter negotiation is not modelled, see below). And each
203 // endpoint's name check is folded into its certificate check, so a
204 // certificate for the wrong name fails one check earlier than it did and
205 // the endpoint halts in the same place.
206 //
207 // Not modelled: parameter negotiation and its downgrade protection
208 // (there is one cipher suite and one group), HelloRetryRequest and
209 // cookies, the PSK and 0-RTT modes (tls13-0rtt.vp), post-handshake
210 // client authentication, the exporter interface over the exporter secret,
211 // alerts, record padding, and the within-connection sequence-number
212 // checks beyond what the nonce expresses. Key compromise impersonation
213 // (F.1) is not tested either: it needs a handshake run after the
214 // client's key has leaked, and every handshake here runs at phase 0.
215
216 attacker[active]
217
218 principal CA[
219     knows private caKey
220     caPk = PUBKEY(caKey)
221 ]
222

```

```

223 principal Server[
224     knows private sigKey
225     knows public serverName
226     gSigKey = PUBKEY(sigKey)
227 ]
228
229 principal Client[
230     knows private clientSigKey
231     knows private clientId
232     clientPk = PUBKEY(clientSigKey)
233 ]
234
235 // A second certified server. Its signing key is the attacker's from the
236 // start, so it stands for any legitimately certified site the attacker
237 // controls.
238 principal Mallory[
239     knows private mkey
240     knows public mallName
241     gmSigKey = PUBKEY(mkey)
242     leaks mkey
243 ]
244
245 // Enrolment with the CA is out of band and guarded: a CA that certifies
246 // whatever key arrives on an open wire would be an oracle.
247 Server → CA: [gSigKey]
248 Client → CA: [clientPk]
249 Mallory → CA: [gmSigKey]
250
251 principal CA[
252     knows public serverName, mallName
253     knows private clientId
254     // §4.5.1: a certificate binds an identity to a key. The client's
255     // identity is private, since the client's certificate must stay
256     // confidential (F.1), so its certificate reaches it as the CA's
257     // signature alone and the client assembles the message itself.
258     serverCertSig = SIGN(caKey, CONCAT(serverName, gSigKey))
259     clientCertSig = SIGN(caKey, CONCAT(clientId, clientPk))
260     mallCertSig = SIGN(caKey, CONCAT(mallName, gmSigKey))
261 ]
262
263 // The trust anchor is distributed independently (§4.5.1), hence guarded.
264 CA → Server: [caPk], [serverCertSig]
265 CA → Client: [caPk], [clientCertSig]
266 CA → Mallory: [mallCertSig]
267
268 principal Server[
269     serverCert = CONCAT(serverName, gSigKey, serverCertSig)
270 ]
271
272 principal Mallory[
273     mallCert = CONCAT(mallName, gmSigKey, mallCertSig)
274 ]
275
276 // The client's configuration: which site it means to reach. See the
277 // modelling notes above for why this is a certificate and not a name.
278 Server → Client: [serverCert]
279 Mallory → Client: [mallCert]
280
281 principal Client[
282     knows public peerCert

```

```

283     knows public recordHeader, updateNotRequested
284     knows public seq0, seq1, seq2, seq3
285     knows public labelDerived, labelFinished, labelIv
286     knows public labelCHsTraffic, labelSHsTraffic
287     knows public labelCApTraffic, labelSApTraffic
288     knows public labelExpMaster, labelResMaster
289     knows public labelResumption, labelTrafficUpd
290     knows public ctxServerCV, ctxClientCV
291     expectedName, _, _ = SPLIT(peerCert)
292     // ClientHello (§4.2.2): a key share, fresh per connection.
293     generates ce
294     gce = PUBKEY(ce)
295 ]
296
297 Client → Server: gce
298
299 principal Server[
300     // The server knows which client it serves; see the modelling notes.
301     knows private clientId
302     knows public recordHeader, updateNotRequested
303     knows public seq0, seq1, seq2, seq3
304     knows public labelDerived, labelFinished, labelIv
305     knows public labelCHsTraffic, labelSHsTraffic
306     knows public labelCApTraffic, labelSApTraffic
307     knows public labelExpMaster, labelResMaster
308     knows public labelResumption, labelTrafficUpd
309     knows public ctxServerCV, ctxClientCV
310     // ServerHello (§4.2.3).
311     generates se
312     gse = PUBKEY(se)
313     dhe = DH_KEX(gce, se)
314     // §7.1. HKDF(salt, ikm, nil) is HKDF-Extract; HKDF(secret, label,
315     // transcript) is Derive-Secret. With no PSK the early secret is
316     // HKDF-Extract(0, 0), a public value, and secrecy enters with dhe.
317     earlySecret = HKDF(nil, nil, nil)
318     derivedES = HKDF(earlySecret, labelDerived, nil)
319     handshakeSecret = HKDF(derivedES, dhe, nil)
320     transcriptSH = HASH(gce, gse)
321     cHsTraffic = HKDF(handshakeSecret, labelCHsTraffic, transcriptSH)
322     sHsTraffic = HKDF(handshakeSecret, labelSHsTraffic, transcriptSH)
323     // §7.3: the write IVs of both directions' handshake keys.
324     cHsIv = HKDF(cHsTraffic, labelIv, nil)
325     sHsIv = HKDF(sHsTraffic, labelIv, nil)
326     derivedHS = HKDF(handshakeSecret, labelDerived, nil)
327     mainSecret = HKDF(derivedHS, nil, nil)
328     // §4.5.1: the transcript through the Certificate.
329     transcriptCert = HASH(transcriptSH, serverName, gSigKey, serverCertSig)
330     // §4.5.2: CertificateVerify, with the server context string.
331     serverCV = SIGN(sigKey, CONCAT(ctxServerCV, transcriptCert))
332     transcriptCV = HASH(transcriptCert, serverCV)
333     // §4.5.3: Finished under the finished key of the base key.
334     sFinishedKey = HKDF(sHsTraffic, labelFinished, nil)
335     serverFinished = MAC(sFinishedKey, transcriptCV)
336     // Certificate, CertificateVerify and Finished, sealed under the
337     // server handshake traffic secret (§5.2), record 0 under that key.
338     serverFlight = AEAD_ENC(sHsTraffic, HASH(sHsIv, seq0), CONCAT(serverName, gSigKey,
339         serverCertSig, serverCV, serverFinished), recordHeader)
340     transcriptSFin = HASH(transcriptCV, serverFinished)
341     cApTraffic = HKDF(mainSecret, labelCApTraffic, transcriptSFin)
342     sApTraffic = HKDF(mainSecret, labelSApTraffic, transcriptSFin)

```

```

342     cApIv = HKDF(cApTraffic, labelIv, nil)
343     sApIv = HKDF(sApTraffic, labelIv, nil)
344     exporterSecret = HKDF(mainSecret, labelExpMaster, transcriptSFin)
345 ]
346
347 Server → Client: gse, serverFlight
348
349 principal Client[
350     dhe_c = DH_KEX(gse, ce)
351     earlySecret_c = HKDF(nil, nil, nil)
352     derivedES_c = HKDF(earlySecret_c, labelDerived, nil)
353     handshakeSecret_c = HKDF(derivedES_c, dhe_c, nil)
354     transcriptSH_c = HASH(gce, gse)
355     cHsTraffic_c = HKDF(handshakeSecret_c, labelCHsTraffic, transcriptSH_c)
356     sHsTraffic_c = HKDF(handshakeSecret_c, labelSHsTraffic, transcriptSH_c)
357     cHsIv_c = HKDF(cHsTraffic_c, labelIv, nil)
358     sHsIv_c = HKDF(sHsTraffic_c, labelIv, nil)
359     derivedHS_c = HKDF(handshakeSecret_c, labelDerived, nil)
360     mainSecret_c = HKDF(derivedHS_c, nil, nil)
361     serverFlightDec = AEAD_DEC(sHsTraffic_c, HASH(sHsIv_c, seq0), serverFlight, recordHeader)?
362     certName, certKey, certSig, serverCV_c, serverFinished_c = SPLIT(serverFlightDec)
363     // §4.5.1.3: the chain is validated against the trust anchor, for the
364     // name the client meant to reach.
365     _ = SIGNVERIF(caPk, CONCAT(expectedName, certKey), certSig)?
366     transcriptCert_c = HASH(transcriptSH_c, certName, certKey, certSig)
367     // §4.5.2: the signature is checked under the key in the certificate,
368     // not under anything the client knew in advance.
369     _ = SIGNVERIF(certKey, CONCAT(ctxServerCV, transcriptCert_c), serverCV_c)?
370     transcriptCV_c = HASH(transcriptCert_c, serverCV_c)
371     sFinishedKey_c = HKDF(sHsTraffic_c, labelFinished, nil)
372     expectedServerFinished = MAC(sFinishedKey_c, transcriptCV_c)
373     _ = ASSERT(serverFinished_c, expectedServerFinished)?
374     transcriptSFin_c = HASH(transcriptCV_c, serverFinished_c)
375     cApTraffic_c = HKDF(mainSecret_c, labelCApTraffic, transcriptSFin_c)
376     sApTraffic_c = HKDF(mainSecret_c, labelSApTraffic, transcriptSFin_c)
377     cApIv_c = HKDF(cApTraffic_c, labelIv, nil)
378     sApIv_c = HKDF(sApTraffic_c, labelIv, nil)
379     exporterSecret_c = HKDF(mainSecret_c, labelExpMaster, transcriptSFin_c)
380     // The client's authentication block (§4.5, Table 2): Certificate,
381     // CertificateVerify with the client context string over the
382     // transcript through server Finished, and Finished under the client
383     // handshake traffic secret.
384     transcriptCCert = HASH(transcriptSFin_c, clientId, clientPk, clientCertSig)
385     clientCV = SIGN(clientSigKey, CONCAT(ctxClientCV, transcriptCCert))
386     transcriptCCV = HASH(transcriptCCert, clientCV)
387     cFinishedKey_c = HKDF(cHsTraffic_c, labelFinished, nil)
388     clientFinished = MAC(cFinishedKey_c, transcriptCCV)
389     clientFlight = AEAD_ENC(cHsTraffic_c, HASH(cHsIv_c, seq0), CONCAT(clientId, clientPk,
390         clientCertSig, clientCV, clientFinished), recordHeader)
391     // §7.1: the resumption secret covers the transcript through client
392     // Finished.
393     transcriptCFin_c = HASH(transcriptCCV, clientFinished)
394     resumptionSecret_c = HKDF(mainSecret_c, labelResMaster, transcriptCFin_c)
395     generates appMsg1
396     appData1 = AEAD_ENC(cApTraffic_c, HASH(cApIv_c, seq0), appMsg1, recordHeader)
397 ]
398
399 Client → Server: clientFlight, appData1
400
401 principal Server[

```

```

401 clientFlightDec = AEAD_DEC(cHsTraffic, HASH(cHsIv, seq0), clientFlight, recordHeader)?
402 clientId_s, clientPk_s, clientCertSig_s, clientCV_s, clientFinished_s = SPLIT(clientFlightDec)
403 // The chain is validated against the trust anchor, for the client
404 // this server serves.
405 _ = SIGNVERIF(caPk, CONCAT(clientId, clientPk_s), clientCertSig_s)?
406 transcriptCCert_s = HASH(transcriptSFin, clientId_s, clientPk_s, clientCertSig_s)
407 _ = SIGNVERIF(clientPk_s, CONCAT(ctxCliCV, transcriptCCert_s), clientCV_s)?
408 transcriptCCV_s = HASH(transcriptCCert_s, clientCV_s)
409 cFinishedKey_s = HKDF(cHsTraffic, labelFinished, nil)
410 expectedClientFinished = MAC(cFinishedKey_s, transcriptCCV_s)
411 _ = ASSERT(clientFinished_s, expectedClientFinished)?
412 transcriptCFin = HASH(transcriptCCV_s, clientFinished_s)
413 resumptionSecret = HKDF(mainSecret, labelResMaster, transcriptCFin)
414 // §4.5.3: application data from the client is read only once its
415 // Finished has been verified.
416 appMsg1_s = AEAD_DEC(cApTraffic, HASH(cApIv, seq0), appData1, recordHeader)?
417 // §4.7.1: two NewSessionTickets, each with its own nonce and so its
418 // own PSK; the tickets are records 0 and 1 under sApTraffic.
419 generates ticketNonce1, ticketNonce2
420 resPsk1 = HKDF(resumptionSecret, labelResumption, ticketNonce1)
421 resPsk2 = HKDF(resumptionSecret, labelResumption, ticketNonce2)
422 newSessionTicket1 = AEAD_ENC(sApTraffic, HASH(sApIv, seq0), ticketNonce1, recordHeader)
423 newSessionTicket2 = AEAD_ENC(sApTraffic, HASH(sApIv, seq1), ticketNonce2, recordHeader)
424 generates appMsg2
425 appData2 = AEAD_ENC(sApTraffic, HASH(sApIv, seq2), appMsg2, recordHeader)
426 // §4.7.3: KeyUpdate goes out under the old key; §7.2: the next
427 // generation is one HKDF-Expand-Label away, with its own write IV,
428 // and its sequence numbers start over (§5.3).
429 keyUpdate = AEAD_ENC(sApTraffic, HASH(sApIv, seq3), updateNotRequested, recordHeader)
430 sApTraffic1 = HKDF(sApTraffic, labelTrafficUpd, nil)
431 sApIv1 = HKDF(sApTraffic1, labelIv, nil)
432 generates appMsg3
433 appData3 = AEAD_ENC(sApTraffic1, HASH(sApIv1, seq0), appMsg3, recordHeader)
434 ]
435
436 Server → Client: newSessionTicket1, newSessionTicket2, appData2, keyUpdate, appData3
437
438 principal Client[
439   ticketNonce1_c = AEAD_DEC(sApTraffic_c, HASH(sApIv_c, seq0), newSessionTicket1, recordHeader)?
440   ticketNonce2_c = AEAD_DEC(sApTraffic_c, HASH(sApIv_c, seq1), newSessionTicket2, recordHeader)?
441   resPsk1_c = HKDF(resumptionSecret_c, labelResumption, ticketNonce1_c)
442   resPsk2_c = HKDF(resumptionSecret_c, labelResumption, ticketNonce2_c)
443   appMsg2_c = AEAD_DEC(sApTraffic_c, HASH(sApIv_c, seq2), appData2, recordHeader)?
444   keyUpdate_c = AEAD_DEC(sApTraffic_c, HASH(sApIv_c, seq3), keyUpdate, recordHeader)?
445   _ = ASSERT(keyUpdate_c, updateNotRequested)?
446   sApTraffic1_c = HKDF(sApTraffic_c, labelTrafficUpd, nil)
447   sApIv1_c = HKDF(sApTraffic1_c, labelIv, nil)
448   appMsg3_c = AEAD_DEC(sApTraffic1_c, HASH(sApIv1_c, seq0), appData3, recordHeader)?
449 ]
450
451 phase[1]
452
453 // Both long-term keys are compromised after the connection has closed
454 // (F.1, forward secrecy with respect to long-term keys).
455 principal Server[
456   leaks sigKey
457 ]
458
459 principal Client[
460   leaks clientSigKey

```

```

461 ]
462
463 phase[2]
464
465 // Then the server's current sending secret and one ticket's PSK (F.2,
466 // forward secrecy after key change; F.1, independence of tickets).
467 principal Server[
468     leaks sApTraffic1, resPsk1
469 ]
470
471 scenarios[
472     Client[peerCert = serverCert]
473     Client[peerCert = mallCert]
474 ]
475
476 queries[
477     confidentiality? sHsTraffic
478     confidentiality? sHsTraffic[
479         precondition[Server → Client: appData2]
480     ]
481     confidentiality? cHsTraffic_c[
482         precondition[Client → Server: clientFlight]
483     ]
484     confidentiality? sApTraffic[
485         precondition[Server → Client: appData2]
486     ]
487     confidentiality? cApTraffic_c[
488         precondition[Client → Server: clientFlight]
489     ]
490     confidentiality? exporterSecret[
491         precondition[Server → Client: appData2]
492     ]
493     equivalence? sApTraffic, sApTraffic_c[
494         precondition[Server → Client: appData2]
495         precondition[Client → Server: clientFlight]
496     ]
497     confidentiality? appMsg1
498     confidentiality? appMsg2
499     confidentiality? appMsg3
500     confidentiality? resPsk2
501     confidentiality? clientId
502     authentication? Server → Client: serverFlight
503     authentication? Client → Server: clientFlight
504     authentication? Client → Server: appData1
505     authentication? Server → Client: appData2
506     authentication? Server → Client: appData3
507     freshness? sApTraffic
508     freshness? cApTraffic_c
509 ]

```

B On soundness

Verifpal is sound but incomplete. Every attack shown here is a genuine attack on the model as written; a query reported as holding means no attack was found within the search this run performed, which is never a proof that none exists.